Power Query の Text.NewGuid 関数がすべての行で同じ値になるのはなぜなんだい？って話

先般、PBIJP Power Query 秘密特訓「虎の穴」炎の復活編 というイベントをやりまして、そこで話した内容など振り返りつつまとめておこうかなと。

https://twitter.com/PowerBIxyz/status/1373433356419932163

# 拗らせるに至った軌跡のひとつから
Power Query の特徴を理解できそうかなぁというテーマ的なものをつぶやいていて、このあたりのお話がよいかなぁと。

https://twitter.com/PowerBIxyz/status/1372176493959213058

そのままなんだけど、 テーブル に [Text.NewGuid - PowerQuery M | Microsoft Docs](https://docs.microsoft.com/en-us/powerquery-m/text-newguid) を使って、GUID 列を追加したときどのような挙動になるかとーく観察するとよいよって話。で、行ごとでユニークな値になるよう考えていくと Power Query の根っこの話が理解できそうかな？と思ったのである。

```text:PowerQuery
Table.AddColumn( SourceTable, "GUID", each Text.NewGuid() )
```
というコードを記述したとき、得られる結果はすべての行で GUID はすべて同じになるのである。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/193509/2af06706-f8c0-f2ee-024b-7df6cd58cc07.png)
で、すべての行でユニークなGUID列を作りにはどうしたらよいかな？の過程から Power Query のスペックを整理しちゃおうぜって話。

## Power Query エディタに表示される値はすべてプレビューである
Power Querry エディタでの表示と得られたクエリ結果が異なることがあるし、
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/193509/08736bf1-8d19-fe50-aba9-e6f3c43413b2.png)
Power Query エディタでも表示される結果が異なることもある。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/193509/26381e85-31bb-61fb-adbd-77b9955a9146.png)
これらは、Powe Query エディタでの表示はすべて**プレビュー**であるから。

エディタ既定の状態で先頭 1000 行を表示しているわけだし、[適用するステップ]ごと通りに処理が実施されるわけではないので、Power Query エディタで表示される通りの結果が得られることは保証されていないのだ。一致することがほとんどというだけなので、信用できるかどうかではなく、**プレビューを参考にする**という感覚での操作が望ましいかなと。

ボタンポチポチだけで済ませてしまうことは否定しないけれども、そのボタンポチひとつで出力される Power Query のコードはプレビューの範囲を対象としたサンプリング結果から生成されているのだから、考えなしでいると失敗を防ぐことができないか、失敗に気づくことがないのだ。 

## すべての行で同じ値になるのはなぜなんだ
### 評価戦略によるもの
すでに評価されている値は再評価しないという戦略だから。
ちょっと記述を変えてみたり、拗らせてみても結果は変わらない。

```text:記述を変えても結果は変わらない
let
    NewGUID = Text.NewGuid(),
    AddedGuid = Table.AddColumn(
        SourceTable,
        "GUID",
        each
            NewGUID,
        Text.Type
    )
in
    AddedGuid
```
```text:拗らせても結果は変わらない
Table.AddColumn(
    SourceTable,
    "GUID",
    each
        Expression.Evaluate( "Text.NewGuid()", #shared ),
    Text.Type
)
```

# なぜ？どうしたら？で理解しようとする
Power Query の仕様を理解しておけばよいだけなんだけど、わかりにくいわりに説明があっさりなので。
[Evaluation model - PowerQuery M | Microsoft Docs](https://docs.microsoft.com/en-us/powerquery-m/evaluation-model)
ポイントはここ。
> **Lazy and eager evaluation**
List, Record, and Table member expressions, as well as let expressions (See Expressions, values, and let expression), are evaluated using lazy evaluation: they are evaluated when needed. All other expressions are evaluated using eager evaluation: they are evaluated immediately, when encountered during the evaluation process. A good way to think about this is to remember that evaluating a list or record expression will return a list or record value that knows how its list items or record fields need to computed, when requested (by lookup or index operators).

## 遅延評価
書いてあることとにらめっこしててもわからないままだから、ななめ読みであっても確かめながらでよいはずだ。 *list*, *record*, *table* は遅延評価されるよと書いてあるなと。で、ほんとなのか？と。

```text:リストで試してみるが
Table.AddColumn(
    SourceTable,
    "GUID",
    each { Text.NewGuid() }{ 0 }
)
```

*list* で試してみるが結果は変わらず、すべての行で GUID 列はすべて同じ値になる。なぜか。
"{ Text.NewGuid() }" という List.Type の値が遅延評価の対象とはなっておらず、行ごとの処理(Table.AddColumn)で再評価されないから。

```text:これも再評価が行われない
Table.AddColumn(
    SourceTable,
    "GUID",
    each { 0, Text.NewGuid() }{ 1 }
)
```

では、行ごとの処理(Table.AddColumn)で再評価が行われるようにするにはどうしたらよいか。カレント行を参照すればよいのだ。

```text:カレント行を参照する
Table.AddColumn(
    SourceTable,
    "GUID",
    each { Text.NewGuid(), _ }{ 0 }
)
```

行ごとの処理(Table.AddColumn) では "{ Text.NewGuid(), _ }" のリストアイテム "\_" はカレント行を表す Record.Type の値になるけれども、行ごとで個別に評価されなければならないから、"{ Text.NewGuid(), _ }" は評価済みの値になりえず、リストに対しアイテムアクセスや集計が行われるまで評価は遅延となる。 "{ 0 }"(アイテムアクセス)で Text.NewGuid() が初めて評価されることになり、結果として異なる GUIDを返す。リストに含まれるアイテム "\_" は参照されることはないから遅延評価という戦略で評価されていない。

```text:let式でもいいし
Table.AddColumn(
    SourceTable,
    "GUID",
    each let current = _ in Text.NewGuid(),
    Text.Type
)
```

```text:recordでもいい
Table.AddColumn(
    SourceTable,
    "GUID",
    each [ Guid = Text.NewGuid(), Current = _ ][Guid],
    Text.Type
)
```
```text:functionでもいい
Table.AddColumn(
    SourceTable,
    "GUID",
    each Function.Invoke((x)=> Text.NewGuid(), {_}),
    Text.Type
)
```

ね、できたでしょ。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/193509/0eea600c-7601-42ec-88d7-42eda13f2942.png)

# 思ったこと🙄
Number.Random 関数も同じ値を返すよね。
DateTime.LocalNow 関数はだいたいの場合で同じ日付時刻を返すよね。

話すことあらかじめまとめておくことがいいかなぁと思ってはいたのだけど、持ち時間なさそうだったしなという都合の良い解釈でほぼ準備なく正直すまんかった。
# その他
- [Power BI 勉強会 - connpass](https://powerbi.connpass.com/)
- [Power BI Desktop を使い始めるには](http://qiita.com/PowerBIxyz/items/6c4735cb5b5712e70981)
- [Excel で Power Query を使い始めるには](http://qiita.com/PowerBIxyz/items/52e563037b855a099ca9)